Solution to Exercise 2.9.3

We denote the irreducible characters of $G := \mathrm{HS}$ by $\chi_1, \ldots, \chi_{24}$, ordered as in the ATLAS.

```
gap> ct := CharacterTable("HS");;
gap> PermChars(ct,100) = [Sum( Irr(ct){[1,2,3]} )];
true
gap> ratchars := Irr(ct){[1,2,3]};;
gap> ratchars := Set( Tensored( ratchars, ratchars) );;
gap> ratchars := Set( Tensored( ratchars, ratchars) );;
gap> mult1 := Union( List( MatScalarProducts( ct, Irr(ct), ratchars ) ,
>                           y -> Positions( y, 1) ));
[ 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 16, 17, 20, 23 ]
```

χ_1, χ_2, χ_3, being the irreducible constituents of the multiplicity free permutation character of degree 100, are characters of rational representations. Of course products of characters of rational representations are also characters of rational representations. Thus `ratchars` is a set of characters of rational representations.

We have computed the set

$$\texttt{mult1} := \{\ i \in \{1, \ldots, 24\} \mid (\chi_i,\ \psi)_G = 1 \quad \text{for some} \quad \psi \in \texttt{ratchars}\ \}.$$

From Theorem 2.9.19 we know that $\mathrm{m}_{\mathbb{Q}}(\chi_i) = 1$ for $i \in \texttt{mult1}$. In case χ_i is rational valued, it is then by Lemma 2.9.15 a character of a rational representation. So we obtain a larger supply of characters of a rational representations.:

```
gap> ratirr := Filtered( mult1, i -> Irr(ct)[i] in RationalizedMat(Irr(ct)) );
[ 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 13, 16, 17, 20, 23 ]
gap> ratchars := Set( Tensored ( Irr(ct){ratirr}, Irr(ct){ratirr} ) );;
gap> Union( List( MatScalarProducts( ct, Irr(ct), ratchars ) ,
>        y -> Positions( y, 1) )) = [1..24] ;
true
```

Thus every $\chi_i \in \mathrm{Irr}(G)$ occurs with multiplicity one in a character of a rational representation and so has Schur index one.